

c1
c2 c3
c4
c5 c6 c7

Towards a miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University

WEIXI MA, Indiana University

DANIEL P. FRIEDMAN, Indiana University

We describe fairness levels in disjunction and conjunction implementations. Specifically, a disjunction implementation can be fair, almost-fair, or unfair. And a conjunction implementation can be fair or unfair. We compare the fairness level of four search strategies: the standard miniKanren interleaving depth-first search, the balanced interleaving depth-first search, the fair depth-first search, and the standard breadth-first search. The two non-standard depth-first searches are new. And we present a new, more efficient and shorter implementation of the standard breadth-first search. Using quantitative evaluation, we argue that a new depth-first searches is a competitive alternative to the standard one, and that our improved breadth-first search implementation is more efficient than the current one.

ACM Reference Format:

Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. Towards a miniKanren with fair search strategies. 1, 1 (May 2019), 16 pages. <https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

1 INTRODUCTION

miniKanren is a family of relational programming languages. Friedman et al. [2, 3] introduce miniKanren and its implementation in *The Reasoned Schemer* and *The Reasoned Schemer, 2nd Ed* (TRS2). Although originally written in Scheme, miniKanren has been ported to many other languages (e.g. OCaml[5]). Byrd et al. [1] have

^{c1}LKC: submission deadline: Mon 27 May 2019

^{c2}DPF: Throughout the paper we have search spaces and search strategies. I can see leaving in the “search strategies”, but let’s try to remove virtually all of the uses of “search” in front of spaces.

^{c3}LKC: Corrected.

^{c4}LKC: BUT, I am thinking about going further — removing “our improved” from this paper. The first reason is that our BFS is actually more efficient than our improved BFS in most benchmarks (see Table 1), despite its theoretical space extravagance. The second reason is that DFSi, DFSf, and our BFS look nice when sitting together.

^{c5}DPF: Does this mean that the comment about append goes away. I really thought that was so clever. I thought that the way you pointed out the trivial changes was very clear, so that did not bother me at all.

^{c6}LKC: Yes, the comment about append will go away if we remove our improved BFS. Then, let’s don’t remove it, i.e. have two new BFS implementations at the same time.

^{c7}LKC: I think having two new BFSs is a bit confusing. So I brand the new and easy/unimproved one as a “stepping-stone” and only mention it in BFS section.

Authors’ addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

© 2019 Association for Computing Machinery.

XXXX-XXXX/2019/5-ART \$15.00

<https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

demonstrated that miniKanren programs are useful in solving several difficult problems. miniKanren.org contains the seeds of many difficult problems and their solutions.

A subtlety arises when a `conde` contains many clauses: not every clause has an equal chance of contributing to the result. As an example, consider the following relation `repeato` and its invocation.

```
(defrel (repeato x out)
  (conde
    ((≡ `(,x) out))
    ((fresh (res)
      (≡ `(,x . ,res) out)
      (repeato x res))))))
> (run 4 q
  (repeato '* q))
'((*) (* *) (* * *) (* * * *))
```

Next, consider the following disjunction of invoking `repeato` with four different letters.

```
> (run 12 q
  (conde
    ((repeato 'a q))
    ((repeato 'b q))
    ((repeato 'c q))
    ((repeato 'd q))))
```

`conde` intuitively relates its clauses with logical or. And thus an unsuspecting beginner would expect each letter to contribute equally to the result, as follows.

```
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

The `conde` in TRS2, however, generates a less expected result.

```
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

The miniKanren in TRS2 implements interleaving DFS (DFS_i), the cause of this unexpected result. With this search strategy, each `conde` clause takes half of its received computational resources and passes the other half to its following clauses, except for the last clause that takes all resources it receives. In the example above, the a clause takes half of all resources. And the b clause takes a quarter. Thus c and d barely contribute to the result.

^{c8}DPF: their BFS (or shortened to BFS). We need only say this once if we go with “our”, which I don’t like. I like “the new” or “the improved” and “the current”

^{c9}LKC: Corrected. I pick “the improved”.

DFS_i is sometimes powerful for an expert. By carefully organizing the order of cond^e clauses, a miniKanren program can explore more “interesting” clauses than those uninteresting ones, and thus use computational resources efficiently.

DFS_i is not always the best choice. For instance, it might be less desirable for novice miniKanren users—understanding implementation details and fiddling with clause order is not their first priority. There is another reason that miniKanren could use more search strategies than just DFS_i. In many applications, there does not exist one order that serves all purposes. For example, a relational dependent type checker contains clauses for constructors that build data and clauses for eliminators that use data. When the type checker is generating simple and shallow programs, the clauses for constructors had better be at the top of the cond^e expression. When performing proof searches for complicated programs, the clauses for eliminators had better be at the top of the cond^e expression. With DFS_i, these two uses cannot be efficient at the same time. In fact, to make one use efficient, the other one must be more sluggish.

The specification that gives every clause in the same cond^e equal “search priority” is fair disj. And search strategies with almost-fair disj give every clause similar priority. Fair conj, a related concept, is more subtle. We cover it in the next section.

To summarize our contributions, we

- propose and implement **balanced interleaving depth-first search** (DFS_{bi}), a new search strategy with almost-fair disj.
- propose and implement **fair depth-first search** (DFS_f), a new search strategy with fair disj.
- implement in a new way the standard **breadth-first search** (BFS) by Seres et al. [7], a search strategy with fair disj and fair conj. We refer to “BFS” as their implementation and “the improved BFS” as our new one. We formally prove that the two implementations are semantically equivalent, however, the improved BFS runs faster in all benchmarks and is shorter.

c1 c2

2 SEARCH STRATEGIES AND FAIRNESS

In this section, we define fair disj, almost-fair disj and fair conj. Before going further into fairness, we give a short review of the terms: *state*, *space*, and *goal*. A *state* is a collection of constraints. (Here, we restrict constraints to unification constraints.) Every answer corresponds to a state. A *space* is a collection of states. And a *goal* is a function from a state to a space.

Now we elaborate fairness by running more queries about repeat^o. We never use run* in this paper because fairness is more interesting when we have an unbounded number of answers. It is perfectly fine, however, to use run* with any search strategies.

2.1 fair disj

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer list. DFS_i, TRS2’s search strategy, however, results in many more lists of *a* than lists of other letters. And some letters (e.g. *c* and *d*) are rarely seen. The more clauses the worse the situation.

^{c1}DPF: The citetseres1999algebra must be not broken: You can either rewrite the sentence or use a tilde called a tie that means don’t break the line. Sometimes using tilde does not work out elegant enough, so the first idea works best.

^{c2}LKC: I have rewritten the whole item.

```

142 ;; DFSi (unfair disj)
143 > (run 12 q
144   (conde
145     ((repeato 'a q))
146     ((repeato 'b q))
147     ((repeato 'c q))
148     ((repeato 'd q))))
149 '((a) (a a) (b) (a a a)
150   (a a a a) (b b)
151   (a a a a a) (c)
152   (a a a a a a) (b b b)
153   (a a a a a a a) (d))
154
155

```

Under the hood, the cond^e here is allocating computational resources to four trivially different spaces. The unfair disj in DFS_i allocates many more resources to the first space. On the contrary, fair disj would allocate resources evenly to each space.

<pre> 160 ;; DFS_f (fair disj) 161 > (run 12 q 162 (cond^e 163 ((repeat^o 'a q)) 164 ((repeat^o 'b q)) 165 ((repeat^o 'c q)) 166 ((repeat^o 'd q)))) 167 '((a) (b) (c) (d) 168 (a a) (b b) (c c) (d d) 169 (a a a) (b b b) (c c c) (d d d)) 170 171 </pre>	<pre> 160 ;; BFS (fair disj) 161 > (run 12 q 162 (cond^e 163 ((repeat^o 'a q)) 164 ((repeat^o 'b q)) 165 ((repeat^o 'c q)) 166 ((repeat^o 'd q)))) 167 '((a) (b) (c) (d) 168 (a a) (b b) (c c) (d d) 169 (a a a) (b b b) (c c c) (d d d)) 170 171 </pre>
---	---

Running the same program again with almost-fair disj (e.g. DFS_{bi}) gives the same result. Almost-fair, however, is not completely fair, as shown by the following example.

```

175 ;; DFSbi (almost-fair disj)
176 > (run 16 q
177   (conde
178     ((repeato 'a q))
179     ((repeato 'b q))
180     ((repeato 'c q))
181     ((repeato 'd q))
182     ((repeato 'e q))))
183 '((b) (c) (d) (a)
184   (b b) (c c) (d d) (e)
185   (b b b) (c c c) (d d d) (a a)
186   (b b b b) (c c c c) (d d d d) (e e))
187
188

```

DFS_{bi} is fair only when the number of goals is a power of 2, otherwise, it allocates some goals twice as many resources as the others. In the above example, where the cond^e has five clauses, DFS_{bi} allocates more resources to the clauses of b, c, and d.

We end this subsection with precise definitions of all levels of *disj* fairness. Our definition of *fair disj* is slightly more general than the one in Seres et al. [7], which is only for binary disjunction. We generalize it to a multi-arity one.

DEFINITION 2.1 (FAIR *disj*). *A disj is fair if and only if it allocates computational resources evenly to spaces produced by goals in the same disjunction (i.e., clauses in the same cond^e).*

DEFINITION 2.2 (ALMOST-FAIR *disj*). *A disj is almost-fair if and only if it allocates computational resources so evenly to spaces produced by goals in the same disjunction that the maximal ratio of resources is bounded by a constant.*

DEFINITION 2.3 (UNFAIR *disj*). *A disj is unfair if and only if it is not almost-fair.*

2.2 fair conj

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer list. Search strategies with unfair conj: DFS_i, DFS_{bi}, and DFS_f, however, results in many more lists of a than lists of other letters. And some letters are rarely seen. Here again, as the number of clauses grows, the situation worsens.

Although some strategies have a different level of fairness in *disj*, they have the same behavior when there is no call to a relational definition in cond^e clauses (including this case).

<pre>;; DFS_i (unfair conj) > (run 12 q (fresh (x) (cond^e ((≡ 'a x)) ((≡ 'b x)) ((≡ 'c x)) ((≡ 'd x))) (repeat^o x q))) '((a) (a a) (b) (a a a) (a a a a) (b b) (a a a a a) (c) (a a a a a a) (b b b) (a a a a a a a) (d))</pre>	<pre>;; DFS_f (unfair conj) > (run 12 q (fresh (x) (cond^e ((≡ 'a x)) ((≡ 'b x)) ((≡ 'c x)) ((≡ 'd x))) (repeat^o x q))) '((a) (a a) (b) (a a a) (a a a a) (b b) (a a a a a) (c) (a a a a a a) (b b b) (a a a a a a a) (d))</pre>	<pre>;; DFS_{bi} (unfair conj) > (run 12 q (fresh (x) (cond^e ((≡ 'a x)) ((≡ 'b x)) ((≡ 'c x)) ((≡ 'd x))) (repeat^o x q))) '((a) (a a) (c) (a a a) (a a a a) (c c) (a a a a a) (b) (a a a a a a) (c c c) (a a a a a a a) (d))</pre>
---	---	--

Under the hood, the cond^e and the call to repeat^o are connected by conj. The cond^e goal outputs a space including four trivially different states. Applying the next conjunctive goal, $(\text{repeat}^o \ x \ q)$, produces four trivially different spaces. In the examples above, all search strategies allocate more computational resources to the space of a. On the contrary, fair conj would allocate resources evenly to each space. For example,

```

236 ;; BFS (fair conj)
237 > (run 12 q
238   (fresh (x)
239     (conde
240       ((= 'a x))
241       ((= 'b x))
242       ((= 'c x))
243       ((= 'd x)))
244     (repeato x q)))
245 '((a) (b) (c) (d)
246   (a a) (b b) (c c) (d d)
247   (a a a) (b b b) (c c c) (d d d))

```

A more interesting situation is when the first conjunct produces an unbounded number of states. Consider the following example, a naive specification of fair conj might require search strategies to produce all sorts of singleton lists, but there would not be any lists of length two or longer, which makes the strategies incomplete. A search strategy is *complete* if and only if “every correct answer would be discovered after some finite time” [7], otherwise, it is *incomplete*. In the context of miniKanren, a search strategy is complete means that every correct answer has a position in large enough answer lists.

```

255 ;; naively fair conj
256 > (run 6 q
257   (fresh (xs)
258     (conde
259       ((repeato 'a xs))
260       ((repeato 'b xs))
261       (repeato xs q)))
262   '(((a)) ((b))
263     ((a a)) ((b b))
264     ((a a a)) ((b b b)))

```

Our solution requires a search strategy with *fair conj* to organize states in buckets in spaces, where each bucket contains finite states, and to allocate resources evenly among spaces derived from states in the same bucket. It is up to a search strategy designer to decide by what criteria to put states in the same bucket, and how to allocate resources among spaces related to different buckets.

BFS puts states of the same cost in the same bucket, and allocates resources carefully among spaces related to different buckets such that it produces answers in increasing order of cost. The *cost* of an answer is its depth in the search tree (i.e., the number of calls to relational definitions required to find the answer) [7]. In the above examples, the cost of each answer is equal to their lengths because we need to apply *repeat^o* n times to find an answer of length n . In the following example, every answer is a list of a list of symbols, where inner lists in the same outer list are identical. Here the cost of each answer is equal to the length of its inner lists plus the length of its outer list. For example, the cost of $((a) (a))$ is $1 + 2 = 3$.

^{c1} ^{c2}

^{c1} DPF: Can you express exactly how all costs are determined? I see that for this simple program, your notion of cost makes sense, but how does it generalize to all costs.

^{c2} LKC: Corrected.

```

283 ;; BFS (fair conj)
284 > (run 12 q
285   (fresh (xs)
286    (conde
287     ((repeato 'a xs))
288     ((repeato 'b xs)))
289    (repeato xs q)))
290
291 '(((a)) ((b)))
292   ((a) (a)) ((b) (b))
293   ((a a)) ((b b))
294   ((a) (a) (a)) ((b) (b) (b))
295   ((a a) (a a)) ((b b) (b b))
296   ((a a a)) ((b b b)))

```

We end this subsection with precise definitions of all levels of conj fairness.

DEFINITION 2.4 (FAIR conj). A conj is fair if and only if it allocates computational resources evenly to spaces produced from states in the same bucket. A bucket is a finite collection of states. And search strategies with fair conj should represent spaces with possibly unbounded collections of buckets.

DEFINITION 2.5 (UNFAIR conj). A conj is unfair if and only if it is not fair.

3 INTERLEAVING DEPTH-FIRST SEARCH

In this section, we review the implementation of interleaving depth-first search (DFS_i). We focus on parts that are relevant to this paper. TRS2, chapter 10 and the appendix, “Connecting the wires”, provide a comprehensive description of the miniKanren implementation but limited to unification constraints (\equiv). Fig. 1 and Fig. 2 show parts that are later compared with other search strategies. We follow some conventions to name variables: ss name states; gs (possibly with subscript) name goals; variables ending with $^\infty$ name (search) spaces. Fig. 1 shows the implementation of disj. The first function, disj_2 , implements binary disjunction. It applies the two disjunctive goals to the input state s and composes the two resulting spaces with append^∞ . The following syntax definitions say disj is right-associative. Fig. 2 shows the implementation of conj. The first function, conj_2 , implements binary conjunction. It applies the first goal to the input state, then applies the second goal to states in the resulting space. The helper function append-map^∞ applies its input goal to states in its input space and composes the resulting spaces. It reuses append^∞ for space composition. The following syntax definitions say conj is also right-associative.

4 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

Balanced interleaving DFS (DFS_{bi}) has an almost-fair disj and unfair conj. The implementation of DFS_{bi} differs from DFS_i ’s in the disj macro. When there are one or more disjunctive goals, the new disj builds a balanced binary tree whose leaves are the goals and whose nodes are disj_2 s, hence the name of this search strategy. In contrast, the disj in DFS_i constructs the binary tree in a particularly unbalanced form. We list the new disj with its helper in Fig. 3. The new helper, disj_+ , takes two additional ‘arguments’. They accumulate goals to be put in the left and right subtrees. The first clause handles the case where there is only one goal. In this case, the tree is the goal itself. When there are more goals, we partition the list of goals into two sublists of roughly equal lengths and recur on the two sublists. We move goals to the accumulators in the last clause. As we are moving

```

330 #| Goal × Goal → Goal |#
331 (define (disj2 g1 g2)
332   (lambda (s)
333     (append∞ (g1 s) (g2 s))))
334
335 #| Space × Space → Space |#
336 (define (append∞ s∞ t∞)
337   (cond
338     ((null? s∞) t∞)
339     ((pair? s∞)
340      (cons (car s∞)
341            (append∞ (cdr s∞) t∞)))
342     (else (lambda ()
343              (append∞ t∞ (s∞))))))
344
345
346 (define-syntax disj
347   (syntax-rules ()
348     ((disj) (fail))
349     ((disj g0 g ...) (disj+ g0 g ...))))
350
351
352 (define-syntax disj+
353   (syntax-rules ()
354     ((disj+ g) g)
355     ((disj+ g0 g1 g ...) (disj2 g0 (disj+ g1 g ...))))
356
357

```

Fig. 1. implementation of DFS_i (Part I)

two goals each time, there are two base cases: (1) no goal remains; (2) one goal remains. We handle these two new base cases in the second clause and the third clause, respectively.

5 FAIR DEPTH-FIRST SEARCH

Fair DFS (DFS_f) has fair disj and unfair conj. The implementation of DFS_f differs from DFS_i's in disj₂ (Fig. 4). The new disj₂ calls a new and fair version of append[∞]. append[∞]_{fair} immediately calls its helper, loop, with the first argument, s?, set to #t, which indicates that we haven't swapped s[∞] and t[∞]. The swapping happens at the third cond clause in the helper, where s? is updated accordingly. The first two cond clauses essentially copy the cars and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned above, is just for swapping. When the fourth and last clause runs, we know that both s[∞] and t[∞] are ended with a thunk, and that we have swapped them. In this case, we construct a new thunk. The new thunk swaps back the two spaces in the recursive call to loop. This is unnecessary for fairness—we do it to produce answers in a more readable order.


```

377 #| Goal × Goal → Goal |#
378 (define (conj2 g1 g2)
379   (lambda (s)
380     (append-map∞ g2 (g1 s))))
381
382 #| Goal × Space → Space |#
383 (define (append-map∞ g s∞)
384   (cond
385     ((null? s∞) '())
386     ((pair? s∞)
387      (append∞ (g (car s∞))
388                (append-map∞ g (cdr s∞))))
389     (else (lambda ()
390              (append-map∞ g (s∞))))))
391
392
393 (define-syntax conj
394   (syntax-rules ()
395     ((conj) (fail))
396     ((conj g0 g ...) (conj+ g0 g ...))))
397
398
399 (define-syntax conj+
400   (syntax-rules ()
401     ((conj+ g) g)
402     ((conj+ g0 g1 g ...) (conj2 g0 (conj+ g1 g ...))))
403
404

```

Fig. 2. implementation of DFS_i (Part II)

```

424 (define-syntax disj
425   (syntax-rules ()
426     ((disj) fail)
427     ((disj g ...) (disj+ (g ...) () ())))))
428
429 (define-syntax disj+
430   (syntax-rules ()
431     ((disj+ () () g) g)
432     ((disj+ (gl ...) (gr ...))
433      (disj2 (disj+ () () gl ...)
434              (disj+ () () gr ...)))
435     ((disj+ (gl ...) (gr ...) g0)
436      (disj2 (disj+ () () gl ... g0)
437              (disj+ () () gr ...)))
438     ((disj+ (gl ...) (gr ...) g0 g1 g ...)
439      (disj+ (gl ... g0) (gr ... g1) g ...)))
440
441
442

```

Fig. 3. implementation of DFS_{bi}

```

444
445 #| Goal × Goal → Goal |#
446 (define (disj2 g1 g2)
447   (lambda (s)
448     (append∞fair (g1 s) (g2 s))))
449
450 #| Space × Space → Space |#
451 (define (append∞fair s∞ t∞)
452   (let loop ((s? #t) (s∞ s∞) (t∞ t∞))
453     (cond
454       ((null? s∞) t∞)
455       ((pair? s∞)
456        (cons (car s∞)
457              (loop s? (cdr s∞) t∞)))
458       (s? (loop #f t∞ s∞))
459       (else (lambda ()
460                (loop #t (t∞) (s∞)))))))
461
462
463

```

Fig. 4. implementation of DFS_f

6 BREADTH-FIRST SEARCH

BFS has both fair disj and fair conj. Our first BFS implementation serves as a “stepping-stone” toward the improved BFS. It is so similar to DFS_f (not DFS_i) that we only need to apply two changes: (1) rename append-map^∞ to $\text{append-map}^\infty_{fair}$ and (2) replace append^∞ with $\text{append}^\infty_{fair}$ in $\text{append-map}^\infty_{fair}$ ’s body.

c1 c2
 , Vol. 1, No. 1, Article . Publication date: May 2019.
 c3 c4
 c5 c6

^{c1}DPF: I need clarification: Applying the above changes won’t result in “our BFS,” but yet a third implementation of BFS?

```

471 #| Goal  $\times$  Space  $\rightarrow$  Space |#
472 (define (append-map∞fair g s∞)
473   (cond
474     ((null? s∞) '())
475     ((pair? s∞)
476      (append∞fair (g (car s∞))
477                    (append-map∞fair g (cdr s∞))))
478     (else (lambda ()
479              (append-map∞fair g (s∞)))))))
480
481
482
483
484

```

Fig. 5. stepping-stone toward the improved BFS (based on DFS_f)

This implementation can be improved in two ways. First, as mentioned in subsection 2.2, BFS puts answers in buckets and answers of the same cost are in the same bucket. In the above implementation, however, it is not obvious how we manage cost information—the cars of a space have cost 0 (i.e., they are all in the same bucket), and every thunk indicates an increment in cost. It is even more subtle that `append∞fair` and the `append-map∞fair` respects the cost information. Second, `append∞fair` is extravagant in memory usage. It makes $O(n + m)$ new cons cells every time, where n and m are the sizes of the first buckets of two input spaces. DFS_f also has the space extravagance.

In the following subsections, we first describe the improved BFS implementation that manages cost information in a more clear and concise way and is less extravagant in memory usage. Then we compare the improved BFS with BFS.

6.1 improved BFS

We make the cost information more clear by changing the Space type, modifying related function definitions, and introducing a few more functions.

The new type of Space is a pair whose car is a list of answers (the bucket), and whose cdr is either #f or a thunk returning a space. A falsy cdr means the space is obviously finite.

We list functions related to the pure subset in Fig. 6. The first three functions are space constructors. `none` makes an empty space; `unit` makes a space from one answer; and `step` makes a space from a thunk. The remaining functions are as before. We compare these functions with BFS in our proof.

^{c2}LKC: If you still need clarification, please add a new note below. Otherwise, you might delete these two notes.

^{c3}DPF: Why are the two `texttt`'s macros, again? Now, it hardly matters, but it is unfriendly to your other co-authors to have us reading long expressions when a newcommand would suffice. I read the paper with a pen marking it up, then I read the paper on the .tex screen. That's each time I read it.

^{c4}LKC: I am sorry about that. Thank you for telling me! I have added many more macros and have removed many uses of `texttt`.

^{c5}DPF: You know that "bag" is a technical terms. It means that there is a structure where each element of the structure has an associated number, which tells you have many occurrences of that element are in the list. For example, the list (a b a c a d c) could be represented like this: (a³ b¹ c² d¹). If that is not what you mean, you might change the name to bucket.

^{c6}LKC: That is not what I mean. I have changed the name to bucket.

^{c1}LKC: I rewrite the second last sentence in the above paragraph such that it better corresponds to the comment about 'append'.

^{c2}DPF: Font Space like in the code comments.

^{c3}LKC: corrected

```

518 #|  $\rightarrow \text{Space}$  |#
519 (define (none)
520   `(() . #f))
521
522 #|  $\text{State} \rightarrow \text{Space}$  |#
523 (define (unit s)
524   `((,s) . #f))
525
526 #|  $(\rightarrow \text{Space}) \rightarrow \text{Space}$  |#
527 (define (step f)
528   `(() . ,f))
529
530
531 #|  $\text{Space} \times \text{Space} \rightarrow \text{Space}$  |#
532 (define (append∞fair s∞ t∞)
533   (cons (append (car s∞) (car t∞))
534         (let ((t1 (cdr s∞)) (t2 (cdr t∞)))
535           (cond
536             ((not t1) t2)
537             ((not t2) t1)
538             (else (lambda ()
539                     (append∞fair (t1) (t2)))))))
540
541
542 #|  $\text{Goal} \times \text{Space} \rightarrow \text{Space}$  |#
543 (define (append-map∞fair g s∞)
544   (foldr
545     (lambda (s t∞)
546       (append∞fair (g s) t∞))
547     (let ((f (cdr s∞)))
548       (step (and f (lambda () (append-map∞fair g (f))))))
549     (car s∞))
550
551
552 #|  $\text{Maybe Nat} \times \text{Space} \rightarrow [\text{State}]$  |#
553 (define (take∞ n s∞)
554   (let loop ((n n) (vs (car s∞)))
555     (cond
556       ((and n (zero? n)) '())
557       ((pair? vs)
558        (cons (car vs)
559              (loop (and n (sub1 n)) (cdr vs))))
560       (else (let ((f (cdr s∞)))
561               (if f (take∞ n (f)) '()))))))
562
563
564

```

Fig. 6. New and changed functions in the improved BFS that implements pure features

```

565 #| Space × (State × Space → Space) × (→ Space) → Space |#
566 (define (elim s∞ fk sk)
567   (let ((ss (car s∞)) (f (cdr s∞)))
568     (cond
569       ((pair? ss) (sk (car ss) (cons (cdr ss) f)))
570       (f (step (lambda () (elim (f) fk sk))))
571       (else (fk)))))
572
573
574 #| Goal × Goal × Goal → Goal |#
575 (define (ifte g1 g2 g3)
576   (lambda (s)
577     (elim (g1 s)
578           (lambda () (g3 s))
579           (lambda (s0 s∞)
580             (append-map∞fair g2
581               (append∞fair (unit s0) s∞))))))
582
583
584 #| Goal → Goal |#
585 (define (once g)
586   (lambda (s)
587     (elim (g s)
588           (lambda () (none))
589           (lambda (s0 s∞) (unit s0)))))
590
591

```

Fig. 7. New and changed functions in the improved BFS that implement impure features

Luckily, the change in $\text{append}_{\text{fair}}^{\infty}$ also fixes the miserable space extravagance—the use of `append` helps us to reuse the first bucket of t^{∞} .

Kiselyov et al. [4] have demonstrated that a *MonadPlus* hides in implementations of logic programming systems. the improved BFS is not an exception: $\text{append-map}_{\text{fair}}^{\infty}$ is like `bind`, but takes arguments in reversed order; `none`, `unit`, and $\text{append}_{\text{fair}}^{\infty}$ correspond to `mzero`, `unit`, and `mplus`, respectively.

c1 c2

Functions implementing impure features are in Fig. 7. The first function, `elim`, takes a space s^{∞} and two continuations `fk` and `sk`. When s^{∞} contains no answers, it calls `fk` with no argument. Otherwise, it calls `sk` with the first answer and the rest of the space. This function is similar to an eliminator of spaces, hence the name. The remaining functions are as before.

6.2 compare the improved BFS with BFS

In this subsection, we compare the pure subset of the improved BFS with BFS. We focus on the pure subset because BFS is designed for a pure relational programming system. We prove in Coq that these two search

^{c1}DPF: $ks \rightarrow sk$ and $kf \rightarrow fk$ and you don't need to say f and s mean success continuation and failure continuation.

^{c2}LKC: corrected

benchmark	size	DFS _i	DFS _{bi}	DFS _f	the improved BFS	BFS
very-recursive ^o	100000	166	103	412	451	1024
	200000	283	146	765	839	1875
	300000	429	346	2085	1809	4408
append ^o	100	17	18	17	17	65
	200	145	137	137	142	121
	300	388	384	387	429	371
revers ^o	10	3	4	3	18	36
	20	35	35	33	3070	3695
	30	329	333	315	75079	107531
quine-1	1	13	14	10	-	-
	2	30	34	25	-	-
	3	41	48	33	-	-
quine-2	1	22	21	12	-	-
	2	51	46	24	-	-
	3	72	76	32	-	-
'(I love you)-1	99	10	37	-	40	96
	198	22	71	-	374	652
	297	24	292	-	1668	3759
'(I love you)-2	99	485	179	-	44	94
	198	808	638	-	414	630
	297	1265	916	-	1651	3691

Table 1. The results of a quantitative evaluation: running times of benchmarks in milliseconds

strategies are semantically equivalent, since (run n ? g) produces the same result in both the improved BFS and BFS. (See supplements for the formal proof.) To compare efficiency, we translate BFS's Haskell code into Racket (See supplements for the translated code). The translation is direct due to the similarity of the two relational programming systems. The translated code is longer than the improved BFS. And it runs slower in all benchmarks. Details about differences in efficiency are in section 7.

7 QUANTITATIVE EVALUATION

In this section, we compare the efficiency of search strategies. A concise description is in Table 1. A hyphen means "running out of 500 MB memory". The first two benchmarks are from TRS2. `reverso` is from Rozplokh and Boulytchev [6]. The next two benchmarks about quine are based on a similar test case in Byrd et al. [1]. We modify the relational interpreters because we don't have disequality constraints (e.g. `absento`). The sibling benchmarks differ in the `conde` clause order of their relational interpreters. The last two benchmarks are about synthesizing expressions that evaluate to `'(I love you)`. They are also based on a similar test case in Byrd et al. [1]. Again, we modify the relational interpreters for the same reason. And the sibling benchmarks differ in the `conde` clause order of their relational interpreters. The first one has elimination rules (i.e. `application`, `car`, and `cdr`) at the end, while the other has them at the beginning. We conjecture that DFS_i would perform badly in the

second case because elimination rules complicate the problem when synthesizing (i.e., our evaluation supports our conjecture.)

^{c1} ^{c2}

In general, only DFS_i and DFS_{bi} constantly perform well. DFS_f is just as efficient in append^0 , revers^0 , and quines. All BFS implementations have obvious overhead in most cases. Among the three variants of DFS, which all have unfair conj , DFS_f is most resistant to clause permutation in quines, followed by DFS_{bi} then DFS_i . DFS_f , however, runs out of memory in '(I love you)s. In contrast, DFS_{bi} still performs more stable than DFS_i . Thus, we consider DFS_{bi} a competitive alternative to DFS_i , the current standard miniKanren search strategy. Among the two implementations of BFS, the improved BFS constantly perform as well or better. It is hard to conclude how disj fairness affect performance based on these statistics. Fair conj , however, imposes considerable overhead constantly except in append^0 . The reason might be that those strategies tend to keep more intermediate answers in the memory.

8 RELATED WORKS

Yang [8] points out that a disjunct complex would be 'fair' if it were a full and balanced tree.

Seres et al. [7] describe a breadth-first search strategy. We present another implementation. Our implementation is semantically equivalent to theirs. But, ours is shorter and performs better in comparison with a straightforward translation of their Haskell code.

Rozplokhas and Boulytchev [6] address the non-commutativity of conjunction, while our work about disj fairness addresses the non-commutativity of disjunction.

9 CONCLUSION

We analyze the definitions of fair disj and fair conj , then propose a new definition of fair conj . Our definition of fair conj is orthogonal with completeness.

We devise two new search strategies (i.e. balanced interleaving DFS (DFS_{bi}) and fair DFS (DFS_f)) and devise a new implementation of BFS. These strategies have different features in fairness: DFS_{bi} has an almost-fair disj and unfair conj . DFS_f has fair disj and unfair conj . BFS has both fair disj and fair conj .

Our quantitative evaluation shows that DFS_{bi} is a competitive alternative to DFS_i , the current miniKanren search strategy, and that the improved BFS is more practical than BFS.

We proof formally that the improved BFS is semantically equivalent to BFS. But, the improved BFS is shorter and performs better in comparison with a straightforward translation of their Haskell code.

We acknowledge that we are making large claims about efficiency given that we have just a few benchmarks, but our intuition is based on sound principles. ...

^{c1} ^{c2}

ACKNOWLEDGMENTS

REFERENCES

- [1] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017).
- [2] Daniel P. Friedman, William E. Byrd, and Oleg Kiselyov. 2005. *The Reasoned Schemer*. The MIT Press.
- [3] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*.

^{c1} DPF: We don't have disequality constraints, so no *absento*. You might just say that.

^{c2} LKC: Corrected.

^{c1} DPF: Finish this paragraph with a clarification and you may place this anywhere in the conclusion. This may require some consultation with Weixi.

^{c2} LKC: I have sent an email to Weixi on the clarification.

- [4] Oleg Kiselyov, Chung-chieh Shan, Daniel P Friedman, and Amr Sabry. 2005. Backtracking, interleaving, and terminating monad transformers:(functional pearl). *ACM SIGPLAN Notices* 40, 9 (2005), 192–203.
- [5] Dmitry Kosarev and Dmitry Boulytchev. 2018. Typed embedding of a relational language in OCaml. *arXiv preprint arXiv:1805.11006* (2018).
- [6] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming*. ACM, 18.
- [7] Silvija Seres, J Michael Spivey, and C. A. R. Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.
- [8] Edward Z. Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue 15* (2010), 11.