

# miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University

WEIXI MA, Indiana University

DANIEL P. FRIEDMAN, Indiana University

The syntax of a programming language should reflect its semantics. When writing a  $\text{cond}^e$  expression in miniKanren, a programmer would expect all clauses share the same chance of being explored, as these clauses are written in parallel. The existing search strategy, interleaving depth-first search ( $\text{DFS}_i$ ), however, prioritizes its clauses by the order how they are written down. Similarly, when a  $\text{cond}^e$  is followed by another goal conjunctively, a programmer would expect states in parallel share the same chance of being explored. Again, the answers by  $\text{DFS}_i$  is different from the expectation. We have devised three new search strategies that have different level of fairness in  $\text{disj}$  and  $\text{conj}$ .

<sup>c1</sup>

## ACM Reference Format:

Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. miniKanren with fair search strategies. 1, 1 (May 2019), 14 pages. <https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

## 1 INTRODUCTION

miniKanren is a family of relational programming languages. Friedman et al. [2] introduce miniKanren and its implementation in *The Reasoned Schemer, 2nd Ed* (TR2). miniKanren programs, especially relational interpreters, have been proven to be useful in solving many problems by Byrd et al. [1].

<sup>c2</sup>

A subtlety arises when a  $\text{cond}^e$  contains many clauses: not every clause has an equal chance to contribute to the result. As an example, consider the following relation `repeato` and its invocation.

```
(defrel (repeato x out)
  (conde
    [(== `(,x) out)]
    [(fresh (res)
      (== `(,x . ,res) out)
      (repeato x res))]))
> (run 4 q
  (repeato '* q))
'((*) (* *) (* * *) (* * * *))
```

<sup>c1</sup>LKC: `disj` and `conj` occurs free in the abstract.

<sup>c2</sup>LKC: Shall we delete the sentence about relational interpreters? It is misleading: it is emphasizing relational interpreters, but we say nothing further about them later.

---

Authors' addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

---

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made for distribution, for profit or commercial advantage, and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from [permissions@acm.org](mailto:permissions@acm.org).

© 2019 Association for Computing Machinery.  
XXXX-XXXX/2019/5-ART \$15.00  
<https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

Next, consider the following disjunction of invoking `repeat` with four different letters.

```
> (run 12 q
   (conde
    ((repeato 'a q))
    ((repeato 'b q))
    ((repeato 'c q))
    ((repeato 'd q))))
```

`conde` intuitively relates its clauses with logical or. And thus an unsuspecting beginner would expect each letter to contribute equally to the result, as follows.

```
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

The `conde` in TR2, however, generates a less expected result.

```
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

The miniKanren in TR2 implements interleaving DFS (`DFSi`), the cause of this unexpected result. With this search strategy, each clause takes half of its received computational resources and pass the other half to its following clauses, except for the last clause that takes all resources it receives. In the example above, the `a` clause takes half of all recourses. And the `b` clause takes a quarter. Thus `c` and `d` barely contribute to the result.

`DFSi` is sometimes powerful for an expert. By carefully organizing the order of `conde` clauses, a miniKanren program can explore more “interesting” clauses than those uninteresting ones, and thus use computational resources efficiently. A little miniKanrenner, however, may beg to differ—understanding implementation details and fiddling with clauses order is not the first priority of a beginner.

There is another reason that miniKanren could use more search strategies than just `DFSi`. In many applications, there does not exist one order that serves for all purposes. For example, a relational dependent type checker contains clauses for constructors that build data and clauses for eliminators that use data. When the type checker is used to generate simple and shallow programs, the clauses of constructors should be put in the front of `conde`. When performing proof searches for complicated programs, the clauses of eliminator should take the focus. With `DFSi`, these two uses cannot be efficient at the same time. In fact, to make one use efficient, the other one must be drastically slow.

The specification that every clause in the same `conde` is given equal “search priority” is called `fair disj`. And search strategies with almost-fair `disj` give every clause similar priority. `Fair conj`, a related concept, is more complicated. We defer it to the next section.

To summarize our contribution, we

- propose and implement balanced interleaving depth-first search (`DFSbi`), a new search strategy with almost-fair `disj`.
- propose and implement fair depth-first search (`DFSf`), a new search strategy with fair `disj`.
- implement in a new way breath-first search (`BFS`), a search strategy with fair `disj` and fair `conj`. And we prove formally that our `BFS` implementation is equivalent to the one by Seres et al. [5]. Our code runs faster in all benchmarks and is simpler.

## 2 SEARCH STRATEGIES AND FAIRNESS

In this section, we define fair `disj`, almost-fair `disj` and fair `conj`. Before going further into fairness, we would like to give a short review about state, search space, and goal, because fairness is defined in terms of them. A *state* is a collection of constraints. Every answer corresponds to a state. A *search space* is a collection of states. And a *goal* is a function from a state to a search space.

Now we elaborate fairness by running more queries about `repeato`.

### 2.1 fair `disj`

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the query result. `DFSi`, the current search strategy, however, results in many more lists of `a` than lists of other letters. And some letters (e.g. `c` and `d`) are rarely seen. The situation would be exacerbated if `conde` contains more clauses.

```
;; DFSi (unfair disj)
> (run 12 q
   (conde
    ((repeato 'a q))
    ((repeato 'b q))
    ((repeato 'c q))
    ((repeato 'd q))))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

Under the hood, the `conde` here is allocating computational resource to four trivially different search space. The unfair `disj` in `DFSi` allocates many more resources to the first search space. On the contrary, fair `disj` would allocate resources evenly to each search space.

<pre>;; DFSf (fair disj) &gt; (run 12 q    (cond<sup>e</sup>     ((repeat<sup>o</sup> 'a q))     ((repeat<sup>o</sup> 'b q))     ((repeat<sup>o</sup> 'c q))     ((repeat<sup>o</sup> 'd q)))) '((a) (b) (c) (d)   (a a) (b b) (c c) (d d)   (a a a) (b b b) (c c c) (d d d))</pre>	<pre>;; BFS (fair disj) &gt; (run 12 q    (cond<sup>e</sup>     ((repeat<sup>o</sup> 'a q))     ((repeat<sup>o</sup> 'b q))     ((repeat<sup>o</sup> 'c q))     ((repeat<sup>o</sup> 'd q)))) '((a) (b) (c) (d)   (a a) (b b) (c c) (d d)   (a a a) (b b b) (c c c) (d d d))</pre>
---	--

Running the same program again with almost-fair `disj` (e.g. `DFSbi`) gives the same result. Almost-fair, however, is not completely fair, as shown by the following example.

```

142 ;; biDFS (almost-fair disj)
143 > (run 16 q
144   (conde
145     ((repeato 'a q))
146     ((repeato 'b q))
147     ((repeato 'c q))
148     ((repeato 'd q))
149     ((repeato 'e q))))
150 '((b) (c) (d) (a)
151   (b b) (c c) (d d) (e)
152   (b b b) (c c c) (d d d) (a a)
153   (b b b b) (c c c c) (d d d d) (e e))
154
155

```

DFS<sub>bi</sub> is fair only when the number of goals is a power of 2, otherwise, some goals would be allocated twice as many resources than the others. In the above example, where the cond<sup>e</sup> has five clauses, the clauses of b, c, and d are allocated more resources.

We end this subsection with precise definitions of all levels of disj fairness. Our definition of *fair disj* is slightly generalize from the one by Seres et al. [5]. Their definition is for binary disjunction. We generalize it to multi-arity one.

**DEFINITION 2.1 (FAIR disj).** *A disj is fair if and only if it allocates computational resource evenly to search spaces produced by goals in the same disjunction (i.e. clauses in the same cond<sup>e</sup>).*

**DEFINITION 2.2 (ALMOST-FAIR disj).** *A disj is almost-fair if and only if it allocates computational resource so evenly to search spaces produced by goals in the same disjunction that the maximal ratio of resources is bounded by a constant.*

**DEFINITION 2.3 (UNFAIR disj).** *A disj is unfair if and only if it is not even almost-fair.*

## 2.2 fair conj

c1

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer. Search strategies with unfair conj (e.g. DFS<sub>i</sub>, DFS<sub>bi</sub>, DFS<sub>f</sub>), however, results in many more lists of as than lists of other letters. And some letters are rarely seen. The situation would be exacerbated if cond<sup>e</sup> contains more clauses.

c2

<sup>c1</sup> LKC: Checked grammer of text before this line

<sup>c2</sup> LKC: Should I add a footnote?

“ Although DFS<sub>i</sub>’s disj is unfair in general, it is fair when there is no call to relational definition in cond<sup>e</sup> clauses (including this case). ”

<pre> 189 ;; DFSi (unfair conj) 190 &gt; (run 12 q 191     (fresh (x) 192       (conde 193         ((= 'a x)) 194         ((= 'b x)) 195         ((= 'c x)) 196         ((= 'd x))) 197       (repeat<sup>o</sup> x q))) 198 '((a) (a a) (b) (a a a) 199   (a a a a) (b b) 200   (a a a a a) (c) 201   (a a a a a a) (b b b) 202   (a a a a a a a) (d)) </pre>	<pre> 189 ;; DFSf (unfair conj) 190 &gt; (run 12 q 191     (fresh (x) 192       (conde 193         ((= 'a x)) 194         ((= 'b x)) 195         ((= 'c x)) 196         ((= 'd x))) 197       (repeat<sup>o</sup> x q))) 198 '((a) (a a) (b) (a a a) 199   (a a a a) (b b) 200   (a a a a a) (c) 201   (a a a a a a) (b b b) 202   (a a a a a a a) (d)) </pre>	<pre> 189 ;; DFSbi (unfair conj) 190 &gt; (run 12 q 191     (fresh (x) 192       (conde 193         ((= 'a x)) 194         ((= 'b x)) 195         ((= 'c x)) 196         ((= 'd x))) 197       (repeat<sup>o</sup> x q))) 198 '((a) (a a) (c) (a a a) 199   (a a a a) (c c) 200   (a a a a a) (b) 201   (a a a a a a) (c c c) 202   (a a a a a a a) (d)) </pre>
--	--	---

Under the hood, the `conde` and the call to `repeato` are connected by `conj`. The `conde` goal outputs a search space including four trivially different states. Then the next conjunctive goal, `(repeato x q)`, is applied to each of these states, producing four trivially different search spaces. In the examples above, the `conjs` are allocating more computational resources to the search space of `a`. On the contrary, fair `conj` would allocate resources evenly to each search space.

```

214 ;; BFS (fair conj)
215 > (run 12 q
216     (fresh (x)
217       (conde
218         ((= 'a x))
219         ((= 'b x))
220         ((= 'c x))
221         ((= 'd x)))
222       (repeato x q)))
223 '((a) (b) (c) (d)
224   (a a) (b b) (c c) (d d)
225   (a a a) (b b b) (c c c) (d d d))

```

A more interesting situation is when the first conjunct produces infinite many answers. Consider the following example, a naive specification of fair `conj` might require search strategies to produce all sorts of singleton lists, but no longer ones, which makes the strategies incomplete. A search strategy is *complete* if and only if it can find out all the answers within finite time, otherwise, it is *incomplete*.

```

236 ;; naively fair conj
237 > (run 6 q
238   (fresh (xs)
239     (conde
240       ((repeato 'a xs))
241       ((repeato 'b xs)))
242     (repeato xs q)))
243 '(((a)) ((b))
244   ((a a)) ((b b))
245   ((a a a)) ((b b b)))

```

Our solution requires a search strategy with *fair conj* to organize states in bags in search spaces, where each bag contains finite states, and to allocate resources evenly among search spaces derived from states in the same bag. It is up to a search strategy designer to decide by what criteria to put states in the same bag, and how to allocate resources among search spaces related to different bags.

BFS puts states of the same cost in the same bag, and allocate resources carefully among search spaces related to different bags such that answers are produced in increasing order of cost. The *cost* of a answer is its depth in the search tree (i.e. the number of calls to relational definitions required to find them) [5]. In the following example, the cost of each answer is equal to the length of the inner lists plus the length of the outer list.

```

256 ;; BFS (fair conj)
257 > (run 12 q
258   (fresh (xs)
259     (conde
260       ((repeato 'a xs))
261       ((repeato 'b xs)))
262     (repeato xs q)))
263 '(((a)) ((b))
264   ((a) (a)) ((b) (b))
265   ((a a)) ((b b))
266   ((a) (a) (a)) ((b) (b) (b))
267   ((a a) (a a)) ((b b) (b b))
268   ((a a a)) ((b b b)))

```

We end this subsection with precise definitions of all levels of conj fairness.

**DEFINITION 2.4 (FAIR conj).** *A conj is fair if and only if it allocates computational resource evenly to search spaces produced from states in the same bag. A bag is a finite collection of state. And search strategies with fair conj should represent search spaces with possibly infinite collection of state.*

**DEFINITION 2.5 (UNFAIR conj).** *A conj is unfair if and only if it is not fair.*

### 3 INTERLEAVING DEPTH-FIRST SEARCH

In this section, we review the implementation of interleaving depth-first search (DFS<sub>i</sub>). This review is for comparison with other search strategies in this paper. Thus we focus on parts that are changed later. TRS2 [2] provides a comprehensive description of the whole miniKanren implementation.

```

283 #| Goal × Goal → Goal |#
284 (define (disj2 g1 g2)
285   (lambda (s)
286     (append∞ (g1 s) (g2 s))))
287
288 #| Space × Space → Space |#
289 (define (append∞ s∞ t∞)
290   (cond
291     ((null? s∞) t∞)
292     ((pair? s∞)
293      (cons (car s∞)
294            (append∞ (cdr s∞) t∞)))
295     (else (lambda ()
296              (append∞ t∞ (s∞))))))
297
298
299 #| Goal × Goal → Goal |#
300 (define (conj2 g1 g2)
301   (lambda (s)
302     (append-map∞ g2 (g1 s))))
303
304
305 #| Goal × Space → Space |#
306 (define (append-map∞ g s∞)
307   (cond
308     ((null? s∞) '())
309     ((pair? s∞)
310      (append∞ (g (car s∞))
311                (append-map∞ g (cdr s∞))))
312     (else (lambda ()
313              (append-map∞ g (s∞))))))
314
315
316 (define-syntax disj
317   (syntax-rules ()
318     ((disj) fail)
319     ((disj g) g)
320     ((disj g0 g ...) (disj2 g0 (disj g ...))))))
321
322 (define-syntax conj
323   (syntax-rules ()
324     ((conj) succeed)
325     ((conj g) g)
326     ((conj g0 g ...) (conj2 g0 (conj g ...))))))
327
328 (define-syntax conde
329   (syntax-rules ()
330     ((conde (g ...) ...)
331      (disj (conj g ...) ...))))

```

Fig. 1. implementation of DFS<sub>i</sub>

```

330 (define-syntax disj
331   (syntax-rules ()
332     ((disj) fail)
333     ((disj g ...) (disj+ (g ...) () ())))))
334
335 (define-syntax disj+
336   (syntax-rules ()
337     ((disj+ (g) () ()) g)
338     ((disj+ () (gl ...) (gr ...))
339      (disj2 (disj+ (gl ...) () ())
340              (disj+ (gr ...) () ())))
341     ((disj+ (g0) (gl ...) (gr ...))
342      (disj2 (disj+ (gl ... g0) () ())
343              (disj+ (gr ...) () ())))
344     ((disj+ (g0 g1 g ...) (gl ...) (gr ...))
345      (disj+ (g ...) (gl ... g0) (gr ... g1))))))
346
347
348
349
350
351
352
353
354
355
356
357
358
359
360
361
362
363
364
365
366
367
368
369
370
371
372
373
374
375
376

```

Fig. 2. implementation of  $\text{DFS}_{bi}$ 

Fig. 1 shows part of the implementation of  $\text{DFS}_i$ . We follow a convention to name variables bound to states with 's', to name variables bound to goals with 'g', and to name variables bound to search spaces with a suffix '-inf'. The first function, `disj2`, implements binary disjunction. `append-inf` is its helper, which composes two disjunctive search spaces. The following function, `conj2`, implements binary conjunction. It applies the *first* goal to the input state, then applies the second goal to states in the resulting search space. The latter process is done with a helper function. `append-map-inf` applies its input goal to states in its input search spaces and compose the resulting search spaces. It reuses `append-inf` for search space composition. The following three definitions introduce syntactic sugars that miniKanren users are more familiar with. The first two definitions say disjunction and disjunction are right-associative. The next and last definition says `conde` relates its clauses disjunctively, and goals in the same clause conjunctively.

#### 4 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

Balanced interleaving DFS ( $\text{DFS}_{bi}$ ) has almost-fair `disj` and unfair `conj`. The implementation of  $\text{DFS}_{bi}$  differs from  $\text{DFS}_i$ 's in the `disj` macro. We list the new `disj` with its helper in Fig. 2. The helper `disj+` builds a balanced binary tree whose leaves are the goals and whose nodes are `disj2`s, hence the name of this search strategy. The first argument to `disj+` is a list of all goals. And the next two arguments accumulate goals in the left and right subtrees. The first clause says that when there is one goal, the tree is the goal itself. When there are more goals, the first argument is partitioned into two sub-lists. The partition is done by repetitively dispatching the first two goals (the last clause), until no goal remains (the second clause) or one goal remains (the third clause). In contrast, the `disj` in  $\text{DFS}_i$  constructs the binary tree with the same collection of leaf nodes but in a particularly unbalanced form.



```

377 #| Goal × Goal → Goal |#
378 (define (disj2 g1 g2)
379   (lambda (s)
380     (append∞fair (g1 s) (g2 s))))
381
382 #| Space × Space → Space |#
383 (define (append∞fair s∞ t∞)
384   (let loop ((s? #t)
385             (s∞ s∞)
386             (t∞ t∞))
387     (cond
388      ((pair? s∞)
389       (cons (car s∞)
390             (loop s? (cdr s∞) t∞)))
391     ((null? s∞) t∞)
392     (s? (loop #f t∞ s∞))
393     (else (lambda ()
394             (append∞fair (t∞) (s∞)))))))
395
396
397
398
399
400
401

```

Fig. 3. implementation of DFS<sub>f</sub>

## 5 FAIR DEPTH-FIRST SEARCH

Fair DFS (DFS<sub>f</sub>) has fair disj and unfair conj. The implementation of DFS<sub>f</sub> differs from DFS<sub>i</sub>'s in disj2 (Fig. 3). disj2 is changed to call a new and fair version of append-inf. append-inf/fair immediately calls its helper, loop, with the first argument, s?, set to #t, which indicates that s-inf and t-inf haven't been swapped. The swapping happens at the third cond clause in the helper, where s? is updated accordingly. The first two cond clauses essentially copy the cars and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned above, is just for swapping. When the fourth and last clause runs, we know that both s-inf and t-inf are ended with a thunk. In this case, a new thunk is constructed. The new thunk calls the driver recursively. Here changing the order of t-inf and s-inf won't hurt the fairness (though it will change the order of answers). We swap them back so that answers are produced in a more natural order.

## 6 BREADTH-FIRST SEARCH

BFS has both fair disj and fair conj. Our implementation is based on DFS<sub>f</sub> (not DFS<sub>i</sub>). To implement BFS based on DFS<sub>f</sub>, we need append-map-inf/fair in addition to append-inf/fair. The only difference between append-map-inf/fair and append-map-inf is that the former calls append-inf/fair instead of append-inf.

The implementation can be improved in two ways. First, as mentioned in section 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. In this implementation, however, it is unclear where this information is recorded. Second, append-inf/fair is extravagant in memory usage. It makes  $O(n + m)$  new cons cells every time, where  $n$  and  $m$  are the "length"s of input search spaces. We address these issues in the first subsection.

Both our BFS and Seres's BFS Seres et al. [5] produce answers in increasing order of cost. So it is interesting to see if they are equivalent. We prove so in Coq. The details are in the second subsection.

## 6.1 optimized BFS

As mentioned in section 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. The cost information is recorded subtly – the cars of a search space have cost 0 (i.e. they are in the same bag), and the costs of answers in thunk are computed recursively then increased by one. It is even more subtle that `append-inf/fair` and the `append-map-inf/fair` respects the cost information. We make these facts more obvious by changing the type of search space, modifying related function definitions, and introducing a few more functions.

The new type is a pair whose `car` is a list of answers (the bag), and whose `cdr` is either a `#f` or a thunk returning a search space. A falsy `cdr` means the search space is obviously finite.

Functions related to the pure subset are listed in Fig. 4 (the others in Fig. 5). They are compared with Seres et al.'s implementation later. The first three functions in Fig. 4 are search space constructors. `none` makes an empty search space; `unit` makes a space from one answer; and `step` makes a space from a thunk. The remaining functions do the same thing as before.

Luckily, the change in `append-inf/fair` also fixes the miserable space extravagance – the use of `append` helps us to reuse the first bag of `t-inf`.

Kiselyov et al. [3] has shown that a *MonadPlus* hides in implementations of logic programming system. Our BFS implementation is not an exception: `append-map-inf/fair` is like `bind`, but takes arguments in reversed order; `none`, `unit`, and `append-inf/fair` correspond to `mzero`, `unit`, and `mplus` respectively.

Functions implementing impure features are in Fig. 5. The first function, `elim`, takes a space `s-inf` and two continuations `ks` and `kf`. When `s-inf` contains some answers, `ks` is called with the first answer and the rest space. Otherwise, `kf` is called with no argument. Here 's' and 'f' means 'succeed' and 'fail' respectively. This function is like an eliminator of search space, hence the name. The remaining functions do the same thing as before.

## 6.2 comparison with the BFS of Seres et al. [5]

In this section, we compare the pure subset of our optimized BFS with the BFS found in Seres et al. [5]. We focus on the pure subset because Silvija's system is pure. Their system represents search spaces with streams of lists of answers, where each list is a bag.

To compare efficiency, we translate her Haskell code into Racket (See supplements for the translated code). The translation is direct due to the similarity in both logic programming systems and search space representations. The translated code is longer and slower than our BFS implementation. Details about difference in efficiency are in section 6.

We prove in Coq that the two BFSs are equivalent, i.e. `(run n g)` produces the same result (See supplements for the formal proof).

## 7 QUANTITATIVE EVALUATION

In this section, we compare the efficiency of search strategies. A concise description is in Table 1. A hyphen means running out of memory. The first two benchmarks are taken from Friedman et al. [2]. `reverso` is from Rozplokh and Boulytchev [4]. Next two benchmarks about quine are modified from a similar test case in Byrd et al. [1]. The modifications are made to circumvent the need for symbolic constraints (e.g. `≠`, `absento`). Our version generates de Bruijnized expressions and prevent closures getting into list. The two benchmarks differ in the `conde` clause order of their relational interpreters. The last two benchmarks are about synthesizing expressions that evaluate to 'I love you'. This benchmark is also inspired by Byrd et al. [1]. Again, the sibling benchmarks differ in the `conde` clause order of their relational interpreters. The first one has elimination rules (i.e. `application`, `car`, and `cdr`) at the end, while the other has them at the beginning. We conjecture that `DFSi` would perform badly in the

```

471 #|  $\rightarrow \text{Space}$  |#
472 (define (none) `(() . #f))
473
474 #|  $\text{State} \rightarrow \text{Space}$  |#
475 (define (unit s) `((,s) . #f))
476
477 #|  $(\rightarrow \text{Space}) \rightarrow \text{Space}$  |#
478 (define (step f) `((() . ,f))
479
480
481 #|  $\text{Space} \times \text{Space} \rightarrow \text{Space}$  |#
482 (define (append∞fair s∞ t∞)
483   (cons (append (car s∞) (car t∞))
484         (let ((t1 (cdr s∞))
485               (t2 (cdr t∞)))
486           (cond
487             ((not t1) t2)
488             ((not t2) t1)
489             (else (lambda () (append∞fair (t1) (t2))))))))
490
491 #|  $\text{Goal} \times \text{Space} \rightarrow \text{Space}$  |#
492 (define (append-map∞fair g s∞)
493   (foldr
494     (lambda (s t∞)
495       (append∞fair (g s) t∞))
496     (let ((f (cdr s∞)))
497       (step (and f (lambda () (append-map∞fair g (f))))))
498     (car s∞))
499
500
501 #|  $\text{Maybe Nat} \times \text{Space} \rightarrow [\text{State}]$  |#
502 (define (take∞ n s∞)
503   (let loop ((n n)
504             (vs (car s∞)))
505     (cond
506       ((and n (zero? n)) '())
507       ((pair? vs)
508        (cons (car vs)
509              (loop (and n (sub1 n)) (cdr vs))))
510       (else
511        (let ((f (cdr s∞)))
512          (if f (take∞ n (f)) '()))))))
513
514
515
516
517

```

Fig. 4. new and changed functions in optimized BFS that implements pure features

```

518 #| Space × (State × Space → Space) × (→ Space) → Space |#
519 (define (elim s∞ ks kf)
520   (let ((ss (car s∞))
521         (f (cdr s∞)))
522     (cond
523       ((and (null? ss) f)
524        (step (lambda () (elim (f) ks kf))))
525       ((null? ss) (kf))
526       (else (ks (car ss) (cons (cdr ss) f))))))
527
528
529 #| Goal × Goal × Goal → Goal |#
530 (define (ifte g1 g2 g3)
531   (lambda (s)
532     (elim (g1 s)
533           (lambda (s0 s∞)
534             (append-map∞fair g2
535              (append∞fair (unit s0) s∞)))
536           (lambda () (g3 s)))))
537
538
539 #| Goal → Goal |#
540 (define (once g)
541   (lambda (s)
542     (elim (g s)
543           (lambda (s0 s∞) (unit s0))
544           (lambda () (none)))))
545
546

```

Fig. 5. new and changed functions in optimized BFS that implements impure features

second case because elimination rules complicate the problem when running backward. The evaluation supports our conjecture.

In general, only  $\text{DFS}_i$  and  $\text{DFS}_{bi}$  constantly perform well.  $\text{DFS}_f$  is just as efficient in all benchmarks but very-recursive. Both BFS have obvious overhead in many cases. Among the three variants of DFS (they all have unfair conj),  $\text{DFS}_f$  is most resistant to clause permutation, followed by  $\text{DFS}_{bi}$  then  $\text{DFS}_i$ . Among the two implementations of BFS, ours constantly performs as well or better. Interestingly, every strategies with fair disj suffers in very-recursive and  $\text{DFS}_f$  performs well elsewhere. Therefore, this benchmark might be a special case. Fair conj imposes overhead constantly except in appendo. The reason might be that strategies with fair conj tend to keep more intermediate answers in the memory.

## 8 RELATED WORKS

Edward points out a disjunct complex would be ‘fair’ if it is a full and balanced tree Yang [6].

Silvija et al Seres et al. [5] also describe a breadth-first search strategy. We proof their BFS is equivalent to ours. But our code looks simpler and performs better in comparison with a straightforward translation of their Haskell code.

benchmark	size	DFS <sub>i</sub>	DFS <sub>bi</sub>	DFS <sub>f</sub>	optimized BFS	Silvija's BFS
very-recursiveo	100000	579	793	2131	1438	3617
	200000	1283	1610	3602	2803	4212
	300000	2160	2836	-	6137	-
appendo	100	31	41	42	31	68
	200	224	222	221	226	218
	300	617	634	593	631	622
reverso	10	5	3	3	38	85
	20	107	98	51	4862	5844
	30	446	442	485	123288	132159
quine-1	1	71	44	69	-	-
	2	127	142	95	-	-
	3	114	114	93	-	-
quine-2	1	147	112	56	-	-
	2	161	123	101	-	-
	3	289	189	104	-	-
'(I love you)-1	99	56	15	22	74	165
	198	53	72	55	47	74
	297	72	90	44	181	365
'(I love you)-2	99	242	61	16	66	99
	198	445	110	60	42	64
	297	476	146	49	186	322

Table 1. The results of a quantitative evaluation: running times of benchmarks in milliseconds

## 9 CONCLUSION

We analysis the definitions of fair disj and fair conj, then propose a new definition of fair conj. Our definition is orthogonal with completeness.

We devise two new search strategies (i.e. balanced interleaving DFS (DFS<sub>bi</sub>) and fair DFS (DFS<sub>f</sub>)) and devise a new implementation of BFS. These strategies have different features in fairness: <sub>bi</sub> has almost-fair disj and unfair conj. DFS<sub>f</sub> has fair disj and unfair conj. BFS has both fair disj and fair conj.

Our quantitative evaluation shows that DFS<sub>bi</sub> and DFS<sub>f</sub> are competitive alternatives to DFS<sub>i</sub>, the current search strategy, and that BFS is less practical.

We prove our BFS is equivalent to the BFS in Seres et al. [5]. Our code is shorter and runs faster than a direct translation of their Haskell code.

## ACKNOWLEDGMENTS

## REFERENCES

- [1] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017).
- [2] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*.
- [3] Oleg Kiselyov, Chung-chieh Shan, Daniel P Friedman, and Amr Sabry. 2005. Backtracking, interleaving, and terminating monad transformers:(functional pearl). *ACM SIGPLAN Notices* 40, 9 (2005), 192–203.
- [4] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming*. ACM, 18.
- [5] Silvija Seres, J Michael Spivey, and C. A. R. Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.

[6] Edward Z. Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue 15* (2010), 11.